

	Application Note
	Product: Domino 1 and 2
Using Domino Utilities Period Function to measure Distance with the T101 Ranging Module	Date: 9/23/98
Introduction: This application note demonstrates Domino's period input utility to measure distance using the sonar ranging module.	
Background: Domino utilities include functions to measure the period of signals applied to either input *INT0 or *INT1. The measurement is based on the time between two falling edges. The sonar ranging module has an open collector output which goes high when a burst of ultrasonics bounces off a solid object and is received by the sonar ranging module. An input to the sonar ranger sends out the burst of ultrasonic. The time between sending and receiving the burst is the distance the burst travels at the speed of sound. The following schematic demonstrates how to connect the Domino to the TI01 ranging module.	
	
How it works: The Domino is used to control the signal used by the TI01 ranging module to initiate an ultrasonic burst and (with a few external components) use the period measurement function to time the burst's return. Domino's floating point math can then easily calculate the distance the burst traveled, based on the period, and output a calculated distance to the object. The distance measurement is formatted in feet and tenth's of a foot. Accuracy is based on target size and distance to target. The echo line is raised when the echoed burst is received. In order to use the period function on DOMINO, the input period to be timed must present 2 falling edges. The time between falling edges is the period (in 1.085uS counts) returned by the function CALL.	



The T1 output begins transmission of the ultrasonic burst. To make the sonar ranger's output look like two falling edges, two things must be done to the ECHO signal. When the ECHO output is inverted the rising edge (signaling the ECHO) becomes a falling edge. This falling edge will stop the period function. We need to create a falling edge to start the period function. To accomplish this a pulse is written out to the INT0 port pin through software. Since the external transistor is an open collector output it can be pulled low by the port itself (see the software listing as to how this period function is tricked into thinking the added pulse comes from the sonar ranger.) There is a slight delay from when the T1 output is set to when the falling edge of the added pulse at INT0 occurs. This is a constant of 0.745ms and must be added to the measured period count.

The total time in ms is:

delay + (period count * .001085)

The total distance the burst traveled in that time is:

Total time in ms * 1.11 feet/ms

The distance to the object creating the echo is:

Total distance/2

The power supply must be sufficient to provide 5 volts at 2A during transmit else 150ma. A VERY large capacitor (>1000uF) soldered right on the sonar ranging module will supply enough current during transmit. Transmission noise can interfere with the ECHO signals edges. The 0.1uF capacitor should squash it. Body capacitance will affect the sonar module, for best results put the module and transducer into a shielded, grounded enclosure (make sure the transducer ground lead goes to its frame and not the silver back.) Remember the minimum distance is about 1.3 feet due to the internal blanking (this can be shortened using the BINH input, see the sonar data sheet.) Have fun

Program Listing:

```

10 MTOP=3FFFH
20 XBY(7000H)=0D2H : XBY(7001H)=0B5H : XBY(7002H)=22H : REM SET T1
30 XBY(7003H)=0C2H : XBY(7004H)=0B5H : XBY(7005H)=22H : REM CLR T1
40 XBY(7006H)=0D2H : XBY(7007H)=0B2H : XBY(7008H)=22H : REM SET INT0
50 XBY(7009H)=0C2H : XBY(700AH)=0B2H : XBY(700BH)=22H : REM CLR INT0
60 CALL 7003H : REM CLR T1
70 PRINT "Hit a key to make a measurement"
80 G=GET : IF G=0 THEN 80
90 CALL 0F070H : REM Watch for falling edge to start measurement
100 CALL 7000H : CALL 7009H : CALL 7006H : REM SET T1, CLR INT0, SET INT0
110 CALL 0F078H : REM check for measurement
120 POP M
130 IF M=0 THEN 110 : REM Waiting for input check again
140 IF M=1 THEN GOTO 110 : REM In process check again
150 IF M=2 THEN PRINT "Error" : GOTO 60 : REM start again
160 M=M+669 : REM Adjustment for delay (SET T1 to CLEAR INT0)

170 M=M*.001085 : REM counts * instruction time
180 M=M*1.11 : REM ms * distance/ms
190 M=M/2 : REM distance/2
200 PRINT USING(##.#),M
210 GOTO 60

```